
1  !  I N T R O D U C T I O N

1.1  !  P U R P O S E

This booklet summarizes the mumps programming language 

for mumps programmers seeking quick reference. It 

summarizes all the variables, operators, functions, commands, 

and other elements of mumps, except for those described in 

other standards (gks, jis90, mwapi, omi, sql, tcp-ip, x3.64, and 

X-Windows).

1. 2  !  A C K N O W L E D G M E N T S

This fifth edition, compiled by Frederick D. S. Marshall, reflects 

the 1995 mumps standard.

The following individuals reviewed and critiqued earlier 

drafts of this guide: Wally Fort, Kristi Hanson, Dave Holbrook, 

Sandra Reynolds, Beverly Marshall Saling, Kate Schell, George 

Timson, Maury Pepper, Larry Landis, Duglas Kilbride, and 

especially David Marcus. Thanks also go to Jack Bowie, Robert 

Greenfield, Dan Schullman, David Sherertz, Robert Stimac, 

George Timson, Tony Wasserman, and Jerry Wilcox, all of 

whom contributed substantially to previous editions.

Special thanks go to Thomas Salander, author of the 

fourth edition (1991); Joel Achtenberg, author of the third and 

second editions (1978 and 1983); and Joan Zimmerman, who 

wrote the original Pocket Guide under grant number hs-01540 

from the National Center for Health Services Research, 

Department of Health, Education, and Welfare.

2  !  O T H E R  R E F E R E N C E S

For an authoritative definition of mumps consult the standard 

itself: X11.1-1995 by the Mumps Development Committee. The 

third edition of the 1995 standard, ISO/IEC 11756:1999, 

Information Technology—Programming Languages—M, is available 

from the International Organization for Standardization (iso) 

and the International Electrotechnical Commission (iec). The 

fourth edition, MUMPS 1995, will be published by Mumps 

Books in 2010.

The seventh and significantly improved edition of the 

mumps pocket guide, MUMPS 1995 Pocket Guide, will be 

published by Mumps Books in 2010.

For an in-depth description of mumps complete with 

extensive examples, consult the MUMPS 1995 Reference 

Manual, which will be published by Mumps Books 2011.

For a self-instruction workbook for students of mumps, 

consult the MUMPS 1995 Workbook, which will be published by 

Mumps Books in 2011.

For a textbook for teaching mumps, consult M 

Programming: A Comprehensive Guide by Richard Walters.

For information about these and other mumps references, 

contact Mumps Books, 819 North 49th Street, Suite 203, 

Seattle, Washington, usa, www.vistaexpertise.net, 

info@vistaexpertise.net, (206) 632-0166.

3  !  T H E  S U I T E  O F  S T A N D A R D S

The mumps programming language is implemented according 

to a standard created and updated by the Mumps 

Development Committee (mdc), which then submits it for 

approval to the American National Standards Institute (ansi), 

the National Institute of Standards and Technology (nist), and 

the International Organization for Standards (iso). The 1977 

and 1984 standards were approved by ansi, the 1990 standard 

was approved by all three bodies, and the 1995 standard 

standard was approved by ansi and iso.

Beginning with the 1995 standard, mumps was defined by 

a suite of standards:

x11.1: The M Programming Language

x11.2: The Open Mumps Interconnect 

x11.3: The Graphical Kernel System

x11.4: The X-Window Binding

x11.6: The M Windowing Api (mwapi)

The mumps standard describes some features by referring 

to other, independent standards, and explaining how to 

access those features in mumps. Such descriptions bind the 

mumps standard to other standards, rather than duplicate the 

work of those standards. The 1995 standard includes six 

bindings: gks, jis90, sql (x3.135), tcp-ip, x3.64, and X-

Windows.

The mumps standard describes most language elements in 

detail; for some it leaves certain portions undefined, left up to 

the implementer. Thus, although use of those features is 

standard, it is not portable across implementations. 

Developers wishing to write code that is completely portable 

across all mumps implementations should avoid the use of 

non-portable features.

Section 2 of the x11.1 standard defines the limits adhered to 

by portable mumps code. Throughout this pocket guide, 

portability guidelines are given within square brackets [like 

this].

Mumps implementers describe those parts of their 

implementations that do not meet the standard in a 

document called the conformance clause.

4



4  !  S Y S T E M  M O D E L

More than just a programming language, mumps provides a 

programming and execution environment, a database, a 

system of interfacing between multiple programming 

languages, and a multi-processing system. The mumps 

programming language is an imperative (command-oriented), 

dynamic, late-binding language oriented toward manipulation 

of strings and sparse arrays with string subscripts.

4 .1  !  M U LT I - P R O C E S S I N G

Mumps systems are designed around multi-processing, in 

which multiple users share the same computing resources. 

Each user of a mumps system is assigned a process with a 

unique id ($job), within which the user can manipulate private 

or shared data. Each process includes a set of structured 

system variables and intrinsic special variables that provide 

the process with information about its characteristics and 

current status, a process stack to aid in executing code, 

storage for the currently executing routine, and a symbol 

table in which to store private data.

Mumps provides ways to equitably share data, 

input/output devices, and other common resources, as well 

as built-in features that the programmer can use to 

synchronize the activities of separate processes (lock), to 

generate new processes (job), or to end the current process 

(halt).

4 . 2  !  D A T A

Each process can create and manipulate its own private data 

in local variables. Unlike most programming languages, 

mumps extends the use of variables to include what other 

systems do with files; mumps systems store shared data in 

global variables. Local variables are temporary, lasting only 

for the duration of the process, but global variables last 

indefinitely, beyond the life of any single process.

Mumps variables need not be declared before use, but are 

instead created dynamically by giving them values. Any 

variable not yet created has an undefined value, and mumps 

provides two special functions for dealing with variables that 

may be undefined ($data and $get). Attempts to retrieve 

directly an undefined value cause an error.

Variables are not statically typed, which would restrict 

them to storing only specific kinds of values; they are 

dynamically typed. Dynamic typing lets variables be of any 

data type at any time, changing types arbitrarily and without 

declaration during execution. Mumps accommodates this by 

treating all data as variable-length strings, but interpreting 

those strings as the data types appropriate to the operations 

being performed upon them. Accordingly, mumps provides 

powerful built-in string manipulation features.

In addition, all variables can be subscripted arrays. These 

arrays are not statically declared, but are dynamically created 

and modified at run time.

Subscripts need not contain contiguous ranges of numeric 

subscripts, but instead can be sparsely populated with only 

the subscripts desired. Subscripts in mumps can have any 

string value, and are automatically collated.

Thus, sorting, which is usually a dominant activity in 

database software, is built into mumps data. Mumps provides 

extensive array and tree manipulation features. As a result, 

mumps data structures are far more dynamic than in most 

languages, performing a significant portion of the computing 

usually reserved for code.

4 . 3  !  C O D E

Processes execute mumps code in modules called routines, 

each of which can execute other routines as needed. 

Additionally, mumps increases the flexibility of the code by 

blurring the distinction between code and data. Any line of 

code can be fetched as data ($text), and assuming its value 

adheres to mumps code syntax, any variable may be executed 

as code (xecute). In addition, mumps routines can leave 

fragments of code unresolved until run time (@, indirection). 

This dynamic interaction between the code and data results in 

highly flexible code whose exact behavior is not bound until 

run time.

4 . 4  !  E N V I R O N M E N T S

Mumps systems collect code and data into environments. As 

each new mumps process is created it is assigned an 

environment, which dictates the capabilities and limitations of 

that process. Some routines and globals are accessible by 

other environments, while others are restricted. In this way 

environments supply both security and modularization to 

complex mumps systems.

4 . 5  !  PA C K A G E S

Mumps processes are not limited to mumps code in their own 

environments. Routines in other environments, or code 

external to mumps systems can be bound up into external 

packages callable by the mumps process, and code written in 

other languages can even be embedded within mumps 

routines.

4 . 6  !  C H A R A C T E R  S E T S

Mumps can operate upon multiple character sets as well, 

reducing the language barrier between computer systems 

and people of different languages. Each character set has its 

own collation system. The 1995 standard includes three 

character set profiles: m, ascii, and jis90.

Although each system has a default character set, the 

system’s processes, routines, globals, and devices can all 

operate under different character sets, so that conceivably 

two people speaking different languages could work on the 

same mumps system in their own language.

4 . 7  !  I N P U T / O U T P U T  D E V I C E S

Each mumps process is assigned a principal input/output 

device ($principal), but can gain ownership of others (open), 

change at any time the current device for input and output 

(use), identify the current device ($io), and release ownership 

of devices (close). Due to the use of a single data type, the 

string, only two commands are needed for input (read) and 

output (write).

The read and write commands share some built-in device 

controls (such as form feed, line feed, and tabbing) for 

formatting output. However, additional device controls are 

control mnemonics bound up into mnemonic spaces. Devices 

can be opened under different mnemonic spaces that dictate 

the control mnemonics available at any time.

5  !  S Y N T A X

The mumps standard uses a metalanguage to describe its 

language features. Each element of this metalanguage is 

defined once in the standard, and then used by name 

thereafter. The following index describes each metalanguage 

element and identifies where in the x11.1 standard it is defined.

5 .1  !  M E T A L A N G U A G E  E L E M E N T  I N D E X

element description, x11.1 page #

: : = metalanguage operator: definition, 6

[ ] metalanguage operator: option, 6

| | metalanguage operator: alternation, 

6

. . . metalanguage operator: repetition, 6

actual actual argument, 47

actuallist actual argument list, 47

actualname actual argument name, 47

algoref algorithm reference, 17

alternation alternation, 42

argument argument of a command, 44

binaryop binary operator, 39

charset character set, 67

charsetexpr character set expression, 17

closeargument close argument, 49

command command, 43

commands commands separated by cs, 8

commandword command word, 43

comment comment, 8

control control character, 7

controlmnemonic control mnemonic, 66

CR carriage return character, 6

cs command separator, 8

element description, x11.1 page #

device device, 18

deviceattribute device attribute, 49

devicekeyword device keyword, 49

deviceparam device parameter, 49

deviceparameters device parameters, 49

devicexpr device expression, 18

digit decimal digit character, 7

dlabel indirect label (evaluated label), 46

doargument do argument, 49

ecode error code, 25

emptystring empty string, 17

entryref entry reference, 46

environment set of distinct names, 15

eoffset error offset, 38

eol end-of-line, 7

eor end-of-routine, 7

exfunc extrinsic function, 23

exp exponent, 21

expr expression, 11

expratom expression atom, 12

expritem expression item, 21

exprtail expression tail, 39

externalroutinename external routine name, 47

externref external reference, 47

extid external identifier, 11

extsyntax external syntax, 11

exttext external text, 11

exvar extrinsic variable, 23

fncodatom $fnumber code atom, 31

fncode $fnumbercode, 30

fncodexpr $fnumber code expression, 30

fncodp $fnumber code p, 31

element description, x11.1 page #

fncodt $fnumber code t, 31

FF form feed character, 6

formalline formal line (line with formal list), 8

formallist formal argument list, 8

format i/o format code, 65

forparameter for argument, 51

function intrinsic function, 29

glvn global or local variable name, 12

gnamind global name indirection, 15

gotoargument goto argument, 52

graphic graphic character (character with 

visible representation), 7

gvn global variable name, 15

gvnexpr global variable name expression, 18

hangargument hang argument, 53

ident identification character, 7

ifargument if argument, 53

intexpr expression, value interpreted as an 

integer, 23

intlit integer literal, 21

jobargument job argument, 53

jobparameters job parameters, 53

killargument kill argument, 54

L metalanguage operator: list (list of), 6

label label of a line, 8

labelref label reference, 46

leftexpr left expression, 61

leftrestricted left restricted, 61

levelline level line (line without formal list), 8

LF line feed character, 6

li level indicator, 8

line line in routine, 7

element description, x11.1 page #

linebody line body, 8

lineref line reference, 46

lname local name, 54

lnamind local name indirection, 12

lockargument lock argument, 54

logicalop logical operator, 41

ls label separator, 8

lvn local variable name, 12

mant mantissa, 21

mergeargument merge argument, 56

mnemonicspace mnemonic space, 58

mnemonicspacename mnemonic space name, 58

mnemonicspec mnemonic space specifier, 58

name name, 7

namevalue name value, 28

newargument new argument, 57

newsvn new special variable name, 57

noncomma non-comma, 25

nonquote non-quote (any graphic character not 

equal to quote), 21

nref name reference, 55

numexpr expression, value interpreted 

numerically, 23

numlit numeric literal, 21

openargument open argument, 57

openparameters open parameters, 58

packagename package name, 47

patatom pattern atom, 42

patcode pattern code, 42

patnonY pattern non-y character, 42

patnonYZ pattern non-y-or-z character, 42

patnonZ pattern non-z character, 42

element description, x11.1 page #

pattern pattern, 41

place place, 38

postcond post-conditional, 45

processid process identifier, 19

processparameters process parameters, 53

readargument read argument, 60

readcount read count, 60

relation relational operator, 40

repcount repeat count in pattern atom, 42

restartargument restart argument, 64

rexpratom restricted expression atom, 12

rgvn restricted global variable name, 15

rlvn restricted local variable name, 12

routine routine, 7

routinebody routine body, 7

routinehead routine head, 7

routinename routine name, 7

routineref routine reference, 19

routinexpr routine expression, 19

setargument set argument, 61

setdestination set destination, 61

setev set error variable, 61

setextract set $extract, 61

setleft set left, 61

setpiece set $piece, 61

SP space character, 6

ssvn structured system variable name, 16

stackcode $stack code, 37

stackcodexpr $stack code expression, 37

strlit string literal, 21

sublit subscript literal, 28

subnonquote subscript non-quote character, 28

element description, x11.1 page #

svn special variable name, 24

system system, 20

systemexpr system expression, 20

textarg $text argument, 38

timeout time-out specification, 45

transparameters transaction parameters, 64

truthop truth operator, 40

tsparam tstart parameter, 64

tstartargument tstart argument, 64

tvexpr expression, value interpreted as a 

truth-value, 23

unaryop unary operator, 27

useargument use argument, 65

V metalanguage operator: evaluation, 6

writeargument write argument, 65

xargument execute argument, 66

6  !  R O U T I N E S

6 .1  !  R O U T I N E  S T R U C T U R E

Mumps software is written in routines, each of which consists 

of a head and a body.

A routine head is the name of the routine followed by a 

carriage return and a line feed. The first character of the name 

must be either a letter (upper or lower case) or the “%” 

character; if the name is longer than one character, the 

remaining characters can be letters and digits (e.g., VanGogh, 

picasso3, %, %Hopper, or %42, but not 42, wyeth%, or 2blake). 

[For portability, routine names should be distinguished by their first 

eight characters and should not contain lower-case letters.]

The routine body is one or more lines and a carriage 

return and line feed at the end of the routine. [For portability, 

no routine may be more than 10,000 characters long.]

6 . 2  !  L I N E S

Each line in a mumps routine consists of a line-start indicator 

(one or more space), optionally preceded by a line label, 

followed by zero or more commands with their associated 

arguments, optionally followed by a comment, and the line 

must end with an end-of-line indicator (a carriage return and 

line feed).

Commands are separated by one or more spaces, but may 

have additional spaces between them. The minimum 

separation for an argumentless command is two characters. 

Spacing between the final command and the comment, when 

one occurs, is the same as between commands. If the line 

contains commands but no comment, no spaces can occur 

after the last command.

The first character of a comment is a semi-colon, and the 

remaining characters are not treated as executable mumps 

code.

Line labels must be unique within a routine. A line label 

may be either an integer or have the same syntax as routine 

names have. Two or more lines in one routine starting with 

the same label cause error m57. [For portability, labels, like 

routine names, should be distinguished by their first eight 

characters and should not contain lower-case letters.]

Beyond these rules, all routine lines fall into one of two 

categories, level lines or formal lines, each of which has 

additional characteristics. Most lines are level lines. After the 

line-start indicator, level lines have zero or more level 

indicators, each of which consists of a period followed by 

zero or more spaces. The level of a level line is the number of 

level-indicator dots at the start of the line body plus one. 

Some examples of level lines are:

VANRIJN ; next line has an empty line body

Raphael

TITIAN I PAINTER="Klee" D  Q

KLEE . R DALI

GRECO .   R KAHLO D

12 . . R KLIMT

%5 ..; what about Renoir?

Formal lines are used for parameter passing, and have a 

level of one (that is, no dots). They require the label, which is 

immediately followed by a list of formal parameters (an open 

parenthesis, a list of variable names, and a close parenthesis). 

The variables must be unsubscripted local variables. Having 

any variable named more than once in the same formal 

parameter list causes error m21. Formal lines must be called 

through an extrinsic or parameter passing; trying to do 

otherwise causes error m11. Some examples of formal lines 

are:

COUNT() Q ^ARTIST("COUNT") ; extrinsic variable

MONET(PAINTING) N YEAR D LOOKUP Q YEAR ; exfunc

%ADD(ARTIST) D NEW Q ; for use with DO or JOB

The standard suggests but does not require that the first 

line of a routine have a label equal to the routine name, have 

no commands, and have a comment that lists a contact, the 

routine’s position within the hierarchy of its package, the date 

and time it was last changed, and/or a description. Here are 

some examples (see Annex E on page 101 of x11.1 for details):

KOLLWITZ ;

DEGAS ;ART@VA.GOV,ART,FRENCH,IMPRESS

ESCHER ;ART;19960210;query handler for MCE

[For portability, the length of a routine line, including any label 

and comment, is limited to the MUMPS string-length limit of 255 

characters. The carriage return and line feed at the end of each line 

are not counted in this limit.]

6 . 3  !  L I N E  R E F E R E N C E S

Whether identifying a starting place for execution or a 

location to examine, mumps code can refer to routine lines 

during execution. By controlling the kinds of line references 

allowed in certain operations, mumps restricts dynamic access 

into routines.

Label references can only identify code on lines that have 

labels, and must explicitly name either the label or the 

routine, possibly both. If only the routine is named, it 

implicitly refers to the first line of the routine, which must then 

have a label. If only a label is named, it implicitly refers to the 

current routine. If the process has access to another 

environment, it can include an environment in the routine 

reference (e.g., list^|"baroque"|painter). Referencing a non-

existent environment causes error m26. References to a non-

existent label cause error m13. Do and job commands that 

pass parameters to the invoked code can only use label 

references, as can extrinsic function and variable names (e.g., 

$$count^painter but not $$davinci+3^painter)

Entry references expand on label references in two ways: 

by allowing label and routine names to be variable (through 

the use of indirection: @artist^painter, mchlnglo^@medium, 

or @artist^@medium; see section 7.8.3 below), and by 

allowing the use of positive offsets to refer to lines without 

labels (e.g., @artist+10^@medium, list+1^painter, and 

print+5). Use of a negative offset causes error m12. Offsets so 

large they do not refer to an existing line cause error m13. The 

do and job commands when not passing parameters, and the 

goto command, can all use entry references, letting them 

execute code from any line of any routine.

Text references, used only by the $text function, expand 

on entry references by removing the need to base line 

references on a label, letting $text refer to lines in routines 

without knowing more than the name. Lines can be referred 

to as offsets from labels or as absolute line counts (e.g., +10, 

+1^@medium, or paint+9^medium, paint^medium), and these 

absolute references can be indirected in various ways (e.g., 

+line^@routine, and @textref). Absolute line counts less 

than zero cause error m5.

Code in external packages bound to mumps can be 

referred to by an external reference, which takes the form of 

an & character followed by an external routine name in the 

form either of a name or name^name (e.g., &pi). Optionally, a 

package name and a period may come before the external 

routine name (e.g., &math.pi). The do command can use 

external references to invoke external code. The names of 

external variables and functions are also external references 

(see section 7.5).

[External references are not portable.]

18



6 . 4  !  E X E C U T I O N

Mumps systems usually execute code in blocks of lines, but 

transaction processing and error processing change the rules 

of execution. The Mumps Windowing Api introduces a fourth 

execution mode, event processing, described in that 

standard. All four mumps execution modes depend upon the 

process stack.

6.4. 1  !  The Process Stack

The ability to stack context is essential to the execution of 

modular code. Mumps implements it in three ways: the do 

and xecute commands, references to extrinsic functions and 

variables, and error processing. All three, when they occur 

within a process, push another level on its process stack. The 

collection of information (called a frame) pushed on the stack 

is used to restore processing and state information when the 

new execution level finishes.

All three kinds of frames push onto the stack the 

execution level, and the execution location of the command 

or expression that created the new frame. The frames of 

extrinsics and argumentless do commands also push the 

current value of the $test variable. So, for example, after 

completing an argumentless do, $test regains the value it had 

before the do command. Error frames push information about 

error conditions during error processing.

Variable scoping features of mumps, such as the new 

command, also use the process stack to save information for 

later recovery when the current execution level completes.

6.4.2  !  Block Processing

Under normal conditions mumps processes code in blocks of 

lines at the same line level. Lines with a greater level are not 

part of the current block, and are ignored. 

The execution frames for blocks are pushed on the stack 

in one of two ways. The execution frame at the first level of 

the process stack is pushed when the process is created, 

whether by signing onto a mumps system or through 

execution of a job command. Blocks of code can be nested by 

the do and xecute commands or by references to extrinsic 

variables or functions; nesting pushes the execution frames of 

the new blocks onto the process stack.

[For portability, MUMPS code should not use more than 127 

levels of the process stack.]

When a block’s execution frame is pushed on the stack, 

the block is executed in a simple flow. The lines within each 

block are executed one at a time, beginning with the first line 

in the block and continuing down sequentially. Within each 

line, commands are executed left to right.

Four commands modify the flow of execution within a 

block. If and else conditionally skip execution of the rest of 

the line. For repeats execution of the rest of the line. Goto 

repositions the flow at the start of another line within the 

block.

Execution frames are popped from the process stack 

when their code blocks end. Blocks automatically end after 

execution of the last line in the block, or they can be ended in 

mid-block by the quit command, when used outside the 

scope of a for command.

When an execution frame is popped from the stack, 

execution resumes where it left off, in mid-expression or mid-

line, depending on how the block was called. In the case of 

extrinsics, the block was called during the evaluation of the 

expression that contains the extrinsic variable or function, so 

mumps uses the extrinsic’s returned value to continue 

expression evaluation.

Note that goto commands at line level one can reposition 

the flow of execution to the start of any other line at level 

one, even in other blocks or other routines. Taken to 

extremes, this can result in a flow of execution not particularly 

block-oriented, though the block processing rules of 

execution still apply.

6.4.3 !  Error Codes

Code Meaning

m1 naked indicator undefined

m2 invalid combination with $fnumber code atom p

m3 $random seed less than 1

m4 no true condition in $select

m5 line reference less than zero

m6 undefined local variable

m7 undefined global variable

m8 undefined intrinsic special variable

m9 divide by zero

m10 invalid pattern match range

m11 no parameters passed

m12 invalid line reference (negative offset)

m13 invalid line reference (line not found)

m14 line level not 1

m15 undefined index variable

m16 argumented quit not allowed

m17 argumented quit required

m18 fixed length read not greater than zero

m19 cannot copy a tree or subtree into itself

m20 line must have a formal parameter list

m21 algorithm specification invalid

m22 set or kill to ^$global when data in global

m23 set or kill to ^$job for non-existent job number

m24 change to collation algorithm while subscripted local 

variables defined

Code Meaning

m26 non-existent environment

m27 attempt to rollback a transaction that is not restartable

m28 mathematical function, parameter out of range

m29 set or kill on structured system variable not allowed 

by implementation

m30 reference to global variable with different collating 

sequence within a collating algorithm

m31 control mnemonic used for device without a 

mnemonic space selected

m32 control mnemonic used in user-defined mnemonic 

space which has no associated line

m33 set or kill to ^$routine when routine exists

m35 device does not support mnemonic space

m36 incompatible mnemonic spaces

m37 read from device identified by the empty string

m38 invalid structured system variable subscript

m39 invalid $name argument

m40 call-by-reference in job actual parameter

m41 invalid lock argument within a transaction

m42 invalid quit within a transaction

m43 invalid range value ($x, $y)

m44 invalid command outside of a transaction

m45 invalid goto reference

m57 more than one defining occurrence of label in routine

m58 too few formal parameters

7  !  E X P R E S S I O N S

The simplest expression in mumps is a variable, a string literal, 

a numeric constant, or a function (functions are discussed in 

sections 7.4, 7.5, and 7.7). Examples of each of these four 

types of expression are respectively:

LASTNAME

"Distaso"

7

$L(LASTNAME)

Such simple expressions are called atomic expressions. 

More complicated expressions can be built up by linking a 

number of atomic expressions by means of the arithmetic and 

other types of operators. For example:

SUM/TOTAL

SEX="Female"

"adult"_"hood"

Mumps expressions are not evaluated in the order used by 

most programming languages. Most languages follow 

arithmetic operator precedence rules: first apply roots, logs, 

and exponentiation, second multiplication and division, and 

third addition and subtraction. Although nearly a universal 

standard, these rules do not declare the precedence of 

expressions that include relational or string manipulation 

operators as well as arithmetic.

To extend these rules of precedence to account for all the 

combinations that can arise in mumps expressions would yield 

too complex a system to remember. Accordingly, mumps has 

its own system of expression evaluation:

Rule 1) All mumps expressions are processed from left to 

right. Whereas 3+4*2 in most languages would equal 11, in 

mumps it equals 14.

Rule 2) Mumps precedence is overridden through the use 

of parentheses, which nest expressions. Thus 3+(4*2) equals 11 

in mumps.

These precedence rules apply to all expression elements 

in mumps, not just operators. 3+(4*$e(268))*$t is either 11 or 0, 

depending on the value of $test; intrinsics, extrinsics, and 

externals have exactly the same precedence as any operator.

7.1  !  VA L U E S

Strings are interpreted as other data types by rules that permit 

a unique representation of those data types as strings, and 

rules that dictate how to interpret those strings as the data 

types they represent. 

7. 1 . 1  !  Representation

Mumps represents constant values as one of two kinds of 

literals, either string or numeric.

String literals can represent any data type. A string literal is 

bounded by quotes (e.g., "Mozart") and contains any string 

of printable characters. Embedded quotes are represented as 

two consecutive quotes (e.g., "Beethoven's ""Eroica"" 

Symphony"). Each quote pair represents a single quote in the 

value denoted by the string literal. An empty string is 

represented as exactly two quotes (i.e., "").

[For portability, character strings should not be longer than 255 

characters, and should only include characters from a standard 

character set.]

Numeric literals are a special case of string literals, a 

shorthand for representing numeric, integer, or truth values. A 

numeric literal has a mantissa optionally followed by the letter 

e and a positive or negative integer exponent; the mantissa 

can be either an integer or a real number (e.g., 3.14159, -42, 

6.0225e23, .6e+2, or 5e-1, but not 0.34, or 2.00).

[For portability, numbers should fall within the exclusive 

interval [-1025 , -10-25] or [10-25 , 1025] or be zero, and should rely 

on no more than fifteen significant digits.]

7. 1 .2  !  Interpretation

Because data has only one base type, the string, mumps 

programmers never need to convert their data between 

types. Instead, data is interpreted appropriately as required by 

context. If a string does not equal the standard representation 

of the type it is being interpreted as, it is automatically 

coerced into the correct representation. Mumps programmers 

use this type coercion to transform their values as needed.

Numeric interpretation involves taking the leftmost portion 

of the string that is either exponential (42e0), decimal (-2.718), 

or integer (1865) in form. It produces canonic numbers (e.g., 

the numeric interpretation of "0.34" is .34). If the string does 

not begin with a valid numeric representation, the numeric 

interpretation is zero.

Integer interpretation is formed from the numeric 

interpretation by dropping any fraction.

Truth-value interpretation is false if the numeric 

interpretation is 0; otherwise it is true.

Here are some interpretations of different strings:

String Numeric Integer Truth Value

"810" 810 810 1

"98 POUNDS" 98 98 1

"" 0 0 0

" 35" 0 0 0

"86+9" 86 86 1

"PAGE 10" 0 0 0

"-8.4" -8.4 -8 1

"86E-1" 8.6 8 1

"---9" -9 -9 1

"-0" 0 0 0

7. 2  !  VA R I A B L E S

There are three kinds of variables: local, global, and special.

7.2 .1  !  Local Var iables

Locals are stored in each process’s symbol table, which lists 

the names, subscripts, and values of all locals currently 

defined for that process. Attempting to evaluate an undefined 

variable causes error m6. Their names have the same syntax 

[and portability restrictions] as routine names. Access to locals 

is restricted within a process through the use of parameter 

passing or the new command; locals not explicitly scoped by 

one of these methods are available to all routines executing 

within the process for the life of the local.

[For portability, variable names, like labels, should be 

distinguished by their first eight characters and should not contain 

lower case letters.]

[For portability, the total space used by the local variables for a 

process must not exceed 10,000 characters. The total length of a 

local variable reference with all its subscripts must not exceed the 

string length limit. The values of subscripts are restricted only by 

the possible values of strings.]

7.2.2  !  Global Var iables

Global names have the same syntax [and portability 

restrictions] as locals; global names are always preceded by a 

leading caret (^) symbol, as in ^mta. Attempting to evaluate 

an undefined global variable causes error m7. Access to 

globals is restricted by environment.

A naked global reference is a shorthand syntax for 

specifying a global variable by omitting the variable name and 

possibly some of the subscripts. The first subscript in the 

subscript list of a naked global reference implicitly refers to 

the last subscript level of the most recent global reference. 

Thus, if a reference has been made to ^x(1), a subsequent 

naked reference to ^(2,3) would access the value of ^x(2,3).

Note that most recent global reference includes any 

reference to any global, with only highly specialized 

exceptions (see the $name function and the lock command). 

Properly determining the most recent global reference 

depends upon a sophisticated understanding of how mumps 

commands are executed, so naked references should be used 

with caution.

Under certain special conditions, the naked indicator is 

not defined, and a reference to it will cause error m1. These 

conditions arise after:

1. a process begins, but before the first full global 

reference executes;

2. an unsubscripted full global reference executes;

3. a transaction rolls back;

4. the default global environment changes; or

5. $query operates upon a global.

7.2.3 !  Special Var iables

Mumps provides four kinds of special variables designed to 

give additional information to the process: intrinsic special 

variables and structured system variables (both of which give 

each process identification and status information), extrinsic 

variables (which let mumps programmers add new special 

variables to reflect the needs of their software), and external 

variables (which provide status information related to external 

packages). Attempting to evaluate an undefined special 

variable causes error m8. With few exceptions, special 

variables cannot be modified.

7. 3  !  S T R U C T U R E D  S Y S T E M  VA R I A B L E S

Identifiable by the ^$ characters that begin their names, 

structured system variables are arrays that describe the 

current status of the system. The name of any structured 

system variable may be abbreviated as specified in the 

standard (e.g., ^$c).

Each describes the characteristics of some system entity 

or resource. The first subscript always identifies the entity or 

resource described by the nodes underneath it. Therefore, in 

the list of nodes defined within each variable, only the second 

and subsequent subscripts are shown.

The ^$job nodes that permit assignment of the default 

environment for globals, locks, and routines, are settable. 

Setting them to a non-existent environment causes no error, 

but the attempts to refer to globals, locks, or routines in that 

environment will cause error m26.

7.3. 1  !  List of Structured System Var iables

^$CHARACTER (  character-set-profile name )

Defines the available character sets. [Only M, ASCII, and JIS90 

are portable.]

^$C("M","INPUT","JIS90") => xform to M from JIS90

^$C("M","OUTPUT","JIS90") => xform from M to JIS90

^$C("M","IDENT") => character validation

^$C("M","PATCODE","U") => U pattern code definition

^$C("M","COLLATE") => collation algorithm

^$DEVICE (  device ID )

Defines the available i/o devices. [Device IDs are not portable.]

$O(^$D("")) => first device ID

^$D("SCK$5030","CHARACTER") => device’s character set

^$D("SCK$5030",attribute) => attribute’s current value

^$GLOBAL ( global variable name )

Defines the global directory.

^$G("^DD","CHARACTER") => global’s character set

^$G("^DD","COLLATE") => its collation algorithm

^$JOB (  process ID )

Defines the current processes.

$O(^$J("")) => first process ID

^$J(42,"CHARACTER") => process’s character set

^$J(42,"GLOBAL") => process’s global environment

^$J(42,"LOCK") => process’s lock environment

^$J(42,"ROUTINE") => process’s routine environment

^$LOCK ( name reference )

Defines the currently held locks.

$O(^$L("")) => first locked name reference

$D(^$L("^DD(1)")) => true if ^DD(1) is currently locked

^$ROUTINE (  routine name )

Defines the routine directory.

$D(^$R("DIC")) => true if DIC routine exists

^$R("DIC","CHARACTER") => routine’s character set

^$SYSTEM (  system ID )

Defines the current mumps system.

^$S("LIVE","CHARACTER") => system’s default char set

^$Z .  .  .  (  vendor-specific value )

Defines vendor-specific system entities. [Structured system 

variables whose names begin with ^$Z are not portable.]

^$ZSPECIFIC("nonstandard") => nonportable

For more information on structured system variables, see 

chapter 7.1.3 of MUMPS 1995.

7. 4  !  E X T R I N S I C  VA R I A B L E S  &  F U N C T I O N S

Identifiable by the $$ characters that begin its name, an 

extrinsic variable is a special variable added to the language 

by a mumps programmer. Its name must be a label reference 

to the code that returns the value of the extrinsic variable 

(with an argumented quit command); for example, 

$$now^time might always equal the current time in human-

readable format.

Similarly, an extrinsic function is a function added to the 

language by a mumps programmer, and its name must have 

the same format as that of an extrinsic variable. It takes a list 

of arguments and returns a value based on those arguments. 

The arguments are passed as parameters (see 8.1.4, Parameter 

Passing, below) to the code that computes its value, and the 

computed value is returned with an argumented quit 

command; for example, $$weekday^time("February 14, 1996") 

would return “Wednesday”, and $$weekday^time("February 

16, 1996") would return “Friday”.

7. 5  !  E X T E R N A L  VA R I A B L E S  &  F U N C T I O N S

Identifiable by the $& characters that begin its name, an 

external variable is a special variable added to the language 

by an external package. The rest of the name must be a valid 

external reference, such as $&math.pi.

Similarly, an external function is a function added by an 

external package. An external function name has the same 

syntax as an external variable name, and must be followed by 

the list of arguments, such as $&math.sin(value).

[External variables and functions are not portable.]

7. 6  !  I N T R I N S I C  S P E C I A L  VA R I A B L E S

Identifiable by the $ character that begins its name, an 

intrinsic special variable has a unique name that may be 

abbreviated to its initial letter or letters (e.g., $ecode or $ec but 

not $e). Intrinsic variables modifiable by the new or set 

commands are so indicated below.

7.6. 1  !  List of Intr insic Special Var iables

$DEVICE

$d = Status of current device. Settable. One to three pieces 

separated by commas; if it evaluates to true (1), the device is in 

an error condition.

$ECODE

$ec = List of current error codes surrounded by commas. 

Settable. Mdc errors begin with m, implementor errors with z, 

user errors with u. For example, "" means no errors, but 

",m6," means a reference to an undefined local variable 

occurred, and error processing is now in effect.

$ESTACK

$es = Counts process stack levels. Newable. Newing it also 

sets it to 0.

$ETRAP

$et = Mumps code that will execute if an error occurs (e.g., s 

$et="d debug^error"). Settable & Newable. Newing $etrap 

stacks its current value but leaves it set to that value.

$HOROLOG

$h = Current date and time as "days,seconds". The first 

number counts the days elapsed since December 31, 1840 at 

midnight, and the second number counts the seconds since 

the last midnight (e.g., "56665,31576" was February 22, 1996 at 

8:46:16 a.m.).

$IO

$i = Current i/o device.

$JOB

$j = Current process id. A unique positive integer.

$KEY

$k = The control-sequence that terminated the current 

device’s last Read command. Settable.

$PRINCIPAL

$p = Principal i/o device.

$QUIT

$q = True (equals 1) if the current context was invoked as an 

extrinsic variable or function; false (0) otherwise. When $quit 

is true, the quit from the current block of code must have an 

argument.

$STACK

$st = Current level of the process stack.

$STORAGE

$s = Characters of free space available for use.

$SYSTEM

$sy = Current system id. 

$TEST

$t = Result of previous timeout or argumented if. Boolean. 1 

means previous if or timed action succeeded. 

$TLEVEL

$tl = Current number of nested transactions.

$TRESTART

$tr = Number of times the current transaction has been 

restarted.

$X

$x = Approximate horizontal cursor or carriage position of 

current device. Settable.

$Y

$y = Approximate vertical cursor or carriage position of 

current device. Settable.

$Z. . .

All implementation-specific intrinsic special variable names 

begin with $z. Settable and Newable only if allowed by 

implementor. [Special variables whose names begin with $Z are 

not portable.]

For more information on intrinsic special variables, see 

chapter 7.1.4.10 of MUMPS 1995.

7. 7  !  I N T R I N S I C  F U N C T I O N S

Identifiable by the $ character that begins its name, an 

intrinsic function has a unique name that may be abbreviated 

to its initial letter or letters (e.g., $ascii or $a but not $as). All 

intrinsic functions take a list surrounded by parentheses of 

one or more arguments separated by commas. All the 

arguments are evaluated before the function is executed, 

even if (as in the case of $get when applied to defined 

variables) some arguments are not needed for the function to 

correctly execute.

7.7. 1  !  List of Intr insic Functions

$ASCII

Ascii number corresponding to one character in a string. The 

first argument is the string to examine. The optional second 

argument is the position within the string of the character 

whose ascii code should be returned, and defaults to 1.

$A("Beethoven") => 66

$A("HAYDN",3) => 89

$CHAR

Characters corresponding to a list of ascii values. Each 

argument gives the code for one character (a negative integer 

yields an empty string).

$C(66,-1,65,67,72) => "BACH"

$DATA

Number indicating whether a variable is defined or has nodes. 

The argument is the variable to evaluate.

>K BORN ; $D(BORN) => 0

>S BORN=1797 ; $D(BORN) => 1

>S BORN(1)=1840 ; $D(BORN) => 11

>K BORN S BORN(0)=1 ; $D(BORN) => 10

$EXTRACT

One or more characters from a string. Can be used as 

destination of set command. The first argument specifies the 

source string. The optional second argument specifies the 

position of the first character to return, and defaults to 1. The 

optional third argument specifies the position of the last 

character of the substring to return, and defaults to the value 

of the second argument.

$E("Brahms") => "B"

$E("Handel",5) => "e"

$E("Mozart",4,6) => "art"

$FIND

Position of character following leftmost occurrence of 

substring in a string. The first argument is the string to search. 

The second argument is the substring to find. The optional 

third argument is the character position within the string from 

which to begin the search, and defaults to 1.

$F("SIBELIUS","I") => 3

$F("SIBELIUS","I",3) => 7

$F("SIBELIUS","I",7) => 0

$FNUMBER

Number formatted according to codes. The first argument is 

the number to format. The second argument is a string of 

codes that describe the formatting to perform (see 

examples). Note that the only code that can be combined 

with "P" (or "p") is ","; including it with any of the others 

causes error m2.

$FN(-42,"P") => "(42)"

$FN(42,"P") => " 42 "

$FN(-42,"T") => "42-"

$FN(42000,",") => "42,000"

$FN(42,"+") => "+42"

$FN(-42,"-") => 42

$FN(42000,"p,",2) => " 42,000.00 "

$GET

Value of a variable, or a default value if variable is not defined. 

The first argument is the variable whose value should be 

returned. The optional second argument is the value to return 

if the variable is undefined, and defaults to "".

>K BORN ; $G(BORN) => ""

$G(BORN,"unknown") => "unknown"

>S BORN=1841 ; $G(BORN) => 1841

$G(BORN,"??") => 1841

$G(BORN,$$ABC) => 1841

$JUSTIFY

Right justified string in a field of spaces. The first argument is 

the string to justify. The second argument is the number of 

character positions to use to right justify the string; if the 

length of the string itself exceeds this number, $justify has no 

effect. The optional third argument is the maximum number 

of fractional digits to return, and defaults to the number of 

digits in the first argument.

$J("3.14159",9) => "  3.14159"

$J("3.14159",9,2) => "     3.14"

$LENGTH

Length of a string, measured in characters or pieces. The first 

argument is the string to evaluate. The optional second 

argument is the delimiter to use to partition the string for 

counting; if absent, the length is counted in characters.

$L("Verdi & Wagner") => 14

$L("Verdi & Wagner","&") => 2

$L("Verdi & Wagner"," ") => 3

$NAME

Evaluated name of a variable with some, all, or no subscripts; 

such a string is called a name value. The first argument is the 

name to evaluate. The optional second argument is the 

maximum number of subscripts to return, and defaults to the 

number of subscripts in the first argument. A negative second 

argument causes error m39.

>S YEAR=1860

$NA(BORN(YEAR,7)) => "BORN(1860,7)"

$NA(BORN(YEAR,7),1) => "BORN(1860)"

$NA(BORN(YEAR,7),0) => "BORN"

$ORDER

Next or previous subscript in a specified array. The first 

argument is the name of the array, with its last subscript being 

the one to traverse. The optional second argument 

determines whether the previous (-1) or next (1) subscript 

value should return, and defaults to 1.

>K BORN S (BORN(1810),BORN(1809))=""

$O(BORN("")) => 1809

$O(BORN(1809)) => 1810

$O(BORN(1810)) => ""

$O(BORN(1810),-1) => 1809

$O(BORN(1809),1) => 1810

$PIECE

Partitions a string into pieces based on a delimiter, and returns 

some of those pieces. Can be used as destination of set 

command. The first argument is the string to partition. The 

second argument is the delimiter to use to partition it. The 

optional third argument is the first piece of the string to 

return, and defaults to 1. The optional fourth argument is the 

last piece of the string to return, and evaluates to the third 

argument.

>S B3="Beethoven,Bach,Brahms"

$P(B3,",") => "Beethoven"

$P(B3,",",3) => "Brahms"

$P(B3,",",1,2) => "Beethoven,Bach"

>S $P(B3,",",2)="BRUCKNER"

B3 => "Beethoven,BRUCKNER,Brahms"

$QLENGTH

Number of subscripts in a variable name, passed as a name 

value (see $name). The argument is the name value to 

evaluate.

$QL("BORN(1675,5)") => 2

$QSUBSCRIPT

Specified part (name, environment, or a subscript) of a 

variable name, passed as a name value. The first argument is 

the source name value. The second argument is the part of 

the name value to return: -1 returns its environment (if one is 

present), 0 returns its unsubscripted name, and positive 

integers return the corresponding subscripts (e.g., 2 returns 

the second subscript).

>S NAME="^|""MUSIC""|DOB(1862,1)"

$QS(NAME,-1) => "MUSIC"

$QS(NAME,0) => "^DOB"

$QS(NAME,1) => 1862

$QS(NAME,2) => 1

$QUERY

Next subscripted variable name in array, returned as a name 

value. The argument is the name value of the starting 

subscripted variable.

>K BORN S BORN(1891)=""

>S BORN(1882,3)=""

$Q(BORN) => "BORN(1882,3)"

$Q(@"BORN(1882,3)") => "BORN(1891)"

$Q(BORN(1891)) => ""

$RANDOM

Random integer uniformly distributed over an interval 

between 0 and maximum-1, inclusive. The argument is a 

number one greater than the maximum value to return. An 

argument value less than 1 causes error m3.

$R(9) => a number between 0 and 8

$REVERSE

Characters of a string in reverse order. The argument is the 

string to reverse.

$RE("Brahms") => "smharB"

$RE("level") => "level"

$SELECT

Value corresponding to first true condition of list, evaluated 

left to right. Each argument is a conditional expression, 

followed by a colon, and then the expression whose value 

$select should return if the corresponding condition 

evaluates to true. Arguments after the first with a true 

condition are not evaluated. If all conditions are false, error 

m4 occurs.

>S SYMPHONY=5

$S('$D(SYMPHONY):"?",1:"!") => "!"

>K SYMPHONY

$S('$D(SYMPHONY):"?",1:"!") => "?"

$STACK

Information about how a level of the process stack was 

created, what code is executing at that level, and what errors 

have accumulated there. The first argument is the level of the 

stack to analyze, and returns a value that indicates how the 

execution frame at that level was pushed onto the stack: if 

due to a command, the full upper-case name of the 

command; if due to an extrinsic, the value "$$"; and if due to 

an error, its error code.

If the first argument equals -1, $stack returns the depth of 

the stack. If 0, an implementation-specific value that indicates 

how this process was started.

The optional second argument returns other information 

about a stack level. The value "ecode" returns a list of an error 

codes added at that stack level. "Mcode" returns the current 

line of mumps code (or value, in the case of an xecute 

command) at that stack level. "Place" returns the location of 

the current command at that stack level.

This example shows the values returned by $stack for a 

hypothetical error situation:

$ST(-1) => 3 

$ST(0) => "JOB" 

$ST(1) => "DO"

$ST(2) => "$$"

$ST(3) => ",M6," 

$ST(2,"ECODE") => ",M6," 

$ST(2,"MCODE") => " K OOPS W !,OOPS"

$ST(2,"PLACE") => "FUMBLE+9^FOO +2"

$TEXT

A line of code from a routine. The argument is a reference to 

the line to return. Use of an absolute line count, as in the third 

example, but with a count less than zero causes error m5.

>S LINE="FUMBLE+9^FOO"

$T(@LINE) => "K OOPS W !,OOPS"

$T(+0^FOO) => "FOO"

$T(^FOO) ; first line of FOO

$TRANSLATE

A translation of a string, in which certain characters are 

removed or replaced. The first argument is the string to 

translate. The second argument is a string of the characters to 

remove (or replace if a third argument is present). The 

optional third argument is a string of the characters with 

which to replace the characters of the second argument in the 

first argument. If the second argument is longer than the third, 

the excess characters at the end of the second argument are 

removed, not replaced.

>S MAN="SHOSTAKOVICH"

$TR(MAN,"H") => "SOSTAKOVIC"

$TR(MAN,"HS") => "OTAKOVIC"

$TR(MAN,"O","U") => "SHUSTAKUVICH"

$TR(MAN,"OS","CU") => "CHUCTAKUVICH"

$TR(MAN,"SHO","E") => "EETAKVIC"

$TR(MAN,"AKVSHOT","MUS") => "MUSIC"

>S UP="ABCDEFGHIJKLMNOPQRSTUVQXYZ"

>S LO="abcdefghijklmnopqrstuvwxyz"

$TR(MAN,HI,LO) => "shostakovich"

$VIEW

Implementation-specific information. $View is a standard 

function, but its arguments and returned values are not 

defined by the standard.

$Z. . .

All implementation-specific intrinsic function names begin 

with $z. [Special variables whose names begin with $Z are not 

portable.]

For more information on intrinsic functions, see chapter 7.1.5 

of MUMPS 1995.

7. 8  !  O P E R A T O R S

Mumps operators provide arithmetic, string manipulation, and 

indirection. They are recognizable (though not uniquely, since 

syntactical delimiters in the language are similar) as one or 

more punctuation characters. Remember that mumps operator 

precedence follows expression-evaluation rules (see the start 

of section 7).

7.8. 1  !  List of Oper ators

Arithmetic Unary Operators

+

Numeric coercion.

+"27.3 days" => 27.3

-

Negate numeric coercion.

-"3.14 radians/second" => -3.14

Arithmetic Binary Operators

+

Sum

2.718+"2 above e" => 4.718

-

Difference

2.12-"6.3 eV" => -4.18

*

Product

1.00794*"2 atoms/H2" => 2.01588

/

Division.  Division by zero (using any of the three division 

operators) causes error m9.

144.132/12.011 => 12

\

Integer division

82.8\"29.5 years/orbit" => 2

#

Modulo division. a#b = a - (b*floor(a/b)), where floor(x) = the 

largest integer not greater than x.

42#5 => 2

-42#5 => 3

42#-5 => -3

-42#-5 => -2

**

Exponentiated value. Results producing complex numbers are 

not defined (e.g., -1**.5). [For portability, exponentiation has 

only seven significant digits of accuracy.]

4**2 => 16

4**.5 => 2

4**-1 => .25

4**-2 => .0625

4**-.5 => .5

Arithmetic Relational Operators

<

Less than

1642<1879 => 1

>

Greater than

1452>1564 => 0

String Binary Operators

_

Concatenates

"Feynman: "_1918 => "Feynman: 1918"

String Relational Operators

=

Equals

1969-5=1964 => 1

1967="1967: M" => 0

1966=01966 => 1

1966="01966" => 0

"Lovelace"="Hopper"+2 => 2

"Lovelace"=" Lovelace "+2 => 2

[

Contains (note: by definition all strings contain the empty 

string "")

"Darwin"["win" => 1

]

Follows

"COPERNICUS"]"KEPLER" => 0

]]

Sorts after (its behavior depends on the current character set’s 

collation algorithm; e.g., in m numbers collate numerically, 

whereas in ascii they collate as strings).

1683]]170 => 1 in M collation

1683]]170 => 0 in ASCII collation

"PRIESTLEY"]]"LAVOISIER" => 1

?

Pattern matches (see pattern match section below for details)

"Leakey"?1A => 0

"Boaz"?1.A => 1

"Fossey"?1U1.5L => 1

"Goodall"?.4L.P6C.E => 0

"Piaget"?1"Pi"4.U => 0

"Skinner"?2A1(1"in",78C).2U => 0

"Harlow"?.(1"Har",.6"Mas")1.AP => 1

"Maslow"?.E1"low".CNP => 1

Logical Operators

&

And

"Watson"&"Crick" => 0

"Morgan"&1735 => 0

1838&1839 => 1

-12000&1996 => 1

1859&0 => 0

!

Or

"Jenner"!"Pasteur" => 0

"Hoffman"!1928 => 1

1898!-400 => 1

1867!0 => 1

'

Not

'"Turing" => 1

''"Babbage" => 0

'"Backus"&1957 => 1

'("Wirth"&"Codd") => 1

"Wirth"'&"Codd" => 1

'("ALGOL"!1959) => 0

"ALGOL"'!1959 => 0

"Chen"'="Jacquard" => 1

"Ada Lovelace"'["Love" => 0

"ENIAC"'<"Naur" => 1

Special Operators

@

Indirect (see indirection section below for details)

>S BORN(-300)="EUCLID"

>S NAME="BORN"

@(NAME_"(-300)") => "EUCLID"

@NAME@(-300) => "EUCLID"

>S TAG="FUMBLE",ROUTINE="FOO"

>D @TAG^@ROUTINE

>S ARG="BORN(1856)=""FREUD"""

>S @ARG

>S PATTERN="1U4L"

"Curie"?@PATTERN => 1

For more information on operators, see chapters 7.1.4.11 and 

7.2.1 of MUMPS 1995.

7.8.2  !  Pattern Matching

Perhaps the second most powerful operator in mumps (for the 

most powerful, see 7.8.3, Indirection, below), pattern match 

tests whether a string matches a certain pattern. Pattern 

matches in MUMPS 1995 have most of the capabilities of 

regular expressions, a widely-used standard for characterizing 

strings, but also are capable of operating on strings from any 

character set.

Each pattern consists of a series of pattern atoms, each of 

which must match part of the string (matching an empty 

string within a string is valid). Each pattern atom has two 

parts: a repetition count that declares how many substrings 

are required, and a pattern element that describes the 

substrings.

For example, the pattern 1u4l has two pattern atoms: 1u 

and 4l. The 1u requires one upper case letter, and the 4l four 

lower case, which is why "Curie"?1U4L evaluate to 1.

The repetition counts can be counts or ranges. Ranges use 

the "." character to specify the range, and can be open at 

either, both, or neither end. For example, 1u, 1.2u, .2u, and 1.u 

all have valid repetition counts. An upper bound less than the 

lower causes error m10.

Pattern elements can be pattern codes. For the m and ascii 

character sets, the available codes are a, c, e, l, n, p, and u, 

matching to alphabetic, control, every character, lower case, 

numeric, punctuation, and upper case, respectively. Pattern 

codes can include any number of these characters in any 

order, such as 1.ap and .cnp.

Pattern elements can also be literal substrings, as in 

"Maslow"?.E1"low".CNP.

Finally, pattern elements can be alternations that contain 

choices of how to satisfy the pattern atom, for example, the 

pattern match "Harlow"?.(1"Har",.6"Mas")1.AP has an 

alternation in its first pattern atom. Whenever an alternation is 

repeated, as is allowed here by the "." repetition count, each 

choice is independent of the previous ones; "", "Har", 

"HarHar", and "MasHar" would all satisfy the first pattern 

atom. In the 1995 standard, the choices within an alternation 

are restricted to pattern atoms, and may not include full 

patterns (although the next mumps standard will permit full 

patterns within alternation).

For flexibility, patterns may be defined at run time through 

indirection, as explained more fully below.

7.8.3 !  Indirection

Indirection, the most powerful operator in mumps, is more an 

underlying capability that spans multiple syntaxes than it is a 

true operator. This is the ability to leave fragments of code 

unresolved; an expression stands in for the code fragments, 

and is evaluated and then resolved into code and executed at 

run time. The advantage of indirection is generalization, the 

capacity to handle classes of problems rather than just 

specific ones. In three out of five cases, the operator or 

delimiter @ identifies the indirection.

In value indirection, common to all programming 

languages, variable names are manipulated rather than 

values, so that the software can manipulate any data that 

satisfies its constraints. The presence of a variable name 

within an expression identifies the value indirection, so no @ 

is needed (e.g., name and ^science("gould") ).

In name indirection the names themselves can vary. Since 

names are used throughout mumps for many different 

purposes, there are many different kinds of name indirection. 

All are identified by the presence of the @ character, and all 

are recursive in multiple ways.

In variable indirection, the name of a local or global 

variable can be specified indirectly:

>S A="Curie",B="A",C=@B S D="@B",E=@D

>S F="B",G=@@F S ^A="Bohr",H=@("^"_B)

>S ^B=@$E(A)

In these examples, a, c, e, g, and ^b are all set to "Curie", 

and ^a and h to "Bohr". In all cases, the expression following 

the @ is evaluated, and the resulting value treated as a new 

variable name; in the case of e, the resulting name itself 

involves indirection, and so begins the resolution of 

indirection anew.

Subscript indirection reduces the need to use 

concatenation when writing code that traverses the subscripts 

of an array specified by indirection:

>S DIAM("EARTH")="12750 km",ROOT="DIAM"

@(ROOT_"(""EARTH"")") => "12750 km"

@ROOT@("EARTH") => "12750 km"

>S ROOT2=$NA(@ROOT@("EARTH"))

>S @ROOT2@("MILES")="7920 mi"

The first use of indirection in this example is variable 

indirection, while the others are subscript indirection. 

Subscript indirection can add subscripts to any level of array 

reference; in the second  and third examples we add an 

"earth" subscript to the name diam, whereas in the fourth 

we add "miles" to diam("earth").

Variable-name indirection is a specialized form of variable 

indirection that permits only unsubscripted locals, and does 

not evaluate the final name to get a data value. It is used by 

the kill, new, and tstart commands, and when passing 

parameters by reference, e.g., d remove^file(.@nobel) and n 

(@sys).

Lock-name indirection is a similar but distinct 

specialization of variable indirection used only for the lock 

command. It permits subscripted or unsubscripted locals or 

globals, and the global references may or may not include an 

environment (e.g., S ADD="^SCIENCE(""Hawking"")" L 

+@ADD ).

Label indirection and routine indirection let a command 

refer to a label or routine at run time dynamically, so the 

command can refer to multiple labels or routines.

Pattern indirection lets the pattern used in a pattern match 

operation vary dynamically at run time, and also uses the @ 

delimiter to identify the indirection. For example:

>S DATA(1)="Democritus",PATTERN(1)="1.A"

>S DATA(2)="430 A.C."

>S PATTERN(2)="1.8N1"" "".1(1"A.D.",1"B.C.")

>S VALUE=1 ; DATA(VALUE)?@PATTERN(VALUE) => 1

>S VALUE=2 ; DATA(VALUE)?@PATTERN(VALUE) => 0

Argument indirection lets the argument of a specific 

command or function vary dynamically at run time, and also 

uses the @ character. Every command that accepts 

arguments, except the for command, will accept a list of one 

or more of its arguments indirectly. For example:

>S IND="COMPUTER=""Mark 1""" S @IND

>S IND="DATE=1944,C=DATE_"":""_COMPUTER" S @IND

Most functions don’t need argument indirection because 

their arguments are of some type that already indirects, such 

as a variable name. Functions that take multiple arguments, 

unlike commands, cannot indirectly refer to more than one 

argument at once; arguments must be indirected individually. 

Even $text, whose argument is not of a type that normally 

accepts complete indirection, can be referred to indirectly.

Only $select violates this pattern; it does not let a 

complete argument, a condition:value pair, be indirected. 

Instead, the condition and value must be indirected 

independently, and the colon delimiter cannot be indirected 

at all.

Command indirection lets the resolution of entire 

commands be deferred until run-time. It is identified not by 

an @ character, but by its invocation with the xecute 

command. See the description of xecute in chapter 8, 

Commands, for details.

>S ACTION="S BORN(1966)=1" X ACTION

8  !  C O M M A N D S

Mumps commands turn the potential of its data structures 

and the evaluation of its expressions into directed activity.

8 .1  !  G E N E R A L  R U L E S

8.1 . 1  !  Basic Syntax

Each command name has an abbreviation that may be used 

instead of the full name. Some commands take arguments; 

others don’t. Those that do are separated from their argument 

by a single space, and can usually accept multiple arguments, 

separated by commas; execution of multiple arguments acts 

like separate instances of the command with a single 

argument each. Command names are case-insensitive (e.g., 

HALT, halt, and hALt are the same command).

8.1 .2  !  Post-condit ionals  (Command/Argument)

Many commands (see definitions in section 5.5) accept post-

conditionals (expressions appended to the command name 

by a colon). Unless a post-conditional evaluates to true, the 

command does not execute or evaluate its arguments 

(therefore, the naked reference is unaffected by the 

command’s argument, for example). Post-conditionals do not 

change $test.

Some branching commands accept post-conditionals on 

their arguments. Unless these conditions evaluate to true, the 

command skips those arguments.

8.1 .3  !  Timeouts

Four commands accept timeouts, which look like argument 

post-conditionals but evaluate instead to the maximum 

number of seconds to wait for the command to succeed. 

Timeouts always set $test to indicate whether the timed 

command succeeded within the allotted time. [For portability, 

the effects of timeouts lasting for non-integer durations, e.g. 2.5, 

should not be relied upon.]

8.1 .4  !  Par ameter  Passing

Extrinsic function references and the do and job commands 

can pass values or variables to the blocks of code they invoke. 

The list of values to pass must follow the name of the function 

or the argument of the do or job command, and must be 

enclosed in parentheses (e.g., $$add^math(length,pause) or 

j print^science("einstein") ). Each value in this list can be an 

arbitrary expression. The list itself is called the actual 

parameter list.

At the receiving end, the referenced label within the 

routine must contain a list of formal parameters at least as 

long as the actual list. Trying to pass more actual parameters 

than there are formal parameters causes error m58. These 

formal parameters must be unsubscripted local variable 

names.

During invocation, when the execution frame of the new 

code block is pushed on the process stack, the actual 

parameters are bound to the formal parameters either by 

value or by reference (note that the job command can only 

pass parameters by value).

When parameters are passed by value, the corresponding 

named formal parameters are first newed (see new command) 

and then set equal to the values passed in the actual list. If the 

actual parameter is also a variable, it is independent of the 

formal parameter; changes to the formal variable do not 

affect the actual parameter. When the block completes and its 

execution frame pops off the stack, the formal parameters are 

also popped, and any pre-existing local variables of the same 

names are restored to their previous states.

When parameters are passed by reference, the 

corresponding named formal parameters are added as new 

names for the same variables. That is, the name-table entries 

for the formal parameters point to the same data-table entries 

that the actual names do. After the block completes and its 

frame pops off the stack, the formal names for the variables 

are unbound from the actual variables and returned to their 

former bindings (if any), but changes to the values of the 

formal parameters within the completed block are reflected in 

the values of the actual parameters.

Whether parameters are passed by value or reference is 

decided by the caller, not the called block of code. The names 

of variables passed by reference are preceded in the actual list 

by a period (e.g., d update^schedule(.patient) ); a period is not 

placed before the name of the corresponding formal 

parameter name within the routine that defines it.

Parameters can be passed to external routines as well, by 

value or reference.

8 . 2  !  L I S T  O F  C O M M A N D S

BREAK

break:postcondition

Stops execution of current process for debugging until 

signaled. [Behavior and arguments not specified by the standard.]

>B

CLOSE

close:postcondition  Device:parameters, . . .

parameters =>

device parameter

 (device parameter:device parameter: . . . )

device parameter =>

vendor-specific expression

device keyword

device attribute=expression

Releases ownership of an i/o device. Close can also use 

device parameters to manipulate the device as it is released. 

If the current device is closed, the special variable $io will be 

empty or reset to a default value. 

[Implementation-specific CLOSE parameters are not portable.]

>C "AUDOUT","VIDOUT"

>C:OFFLINE "SCREEN1":("READY":6)

DO

do:postcondition

do:postcondition  DoArgument:postcondition, . . .

Do Argument =

entry reference

label reference(actual parameters)

external reference(actual parameters)

Executes a subroutine, then returns control to the next 

command after the do; for multiple arguments, each 

subroutine in turn is executed. The line referenced in each 

argument must have a line level of one, or the do will cause 

error m14. Argumentless do executes the following block of 

code, with a line level one greater than the do’s line level, 

then returns.

>D

>D:AUTHOR="SHAKESPEARE"

>D HAMLET,MACBETH,OTHELLO,KINGLEAR

>D ^PLAY:CHARACTR="FALSTAFF",LIST^PLAY

>S WORK="TEMPEST" D @WORK^PLAY

>S WORK="MUCHADO^PLAY" D @WORK

>D ADD^PLAY("LOVE'S LABOUR'S LOST")

ELSE

else

Lets rest of the line execute only if $test evaluates to false.

>E

FOR

for

for  LocalVariable=ForParameters

For Parameters =>

For Parameter,For Parameter, . . .

For Parameter =>

expression

numeric expression:numeric expr.:numeric expr.

Repeats execution of the rest of the line, and sets the value of 

a variable each time. Argumentless for repeats execution of 

the rest of the line without setting a variable. A quit or goto 

command terminates a for loop on the current line; quit 

terminates only the most recent in a series of nested for 

loops on the line, whereas goto terminates all active for 

loops on the line. Argument indirection not allowed.

Each for parameter defines a series of one or more values 

for the variable to accept, and executes the rest of the line 

once for each value in that series. Each for parameter can be 

either a single evaluated expression, or a range of numeric 

values. Ranges include a starting value, an increment with 

which to calculate subsequent values, and an optional 

maximum value. A for loop that runs out of values stops 

without needing a quit or goto.

>F

>F TALE=1:1:53 => executes 53 times

>F TALE=5:2:53 => executes 25 times

5 6



>F TALE=1:1 Q:TALE=53

>F TALE=1:1 G TELL:^NAME(TALE)="PARDONER"

>F TALE="WIFE OF BATH",12:1:13

>F TALE=2:2:8,11:2:21,24,27,28,31

>F AUTHOR="CHAUCER","BOCCACCIO"

>F TALE=1:1:53 F LINE=1:1 Q:'^TEXT(TALE,LINE)

GOTO

goto:postcondition  EntryReference:postcondition, . . .

Transfers execution to a different line of code, without 

returning when that block of code completes. Trying to goto 

a line at a different line level, or trying to cross block 

boundaries when the goto’s line level is greater than one, 

causes error m45.

>G:AUTHOR="DANTE" ^CANON

>G INFERNO:BAD,PARADISO:'BAD

>G VIRGIL^GUIDE:'BOOK3,BEATRICE^GUIDE

>S CANTO=26,BOOK="INFERNO" G @CANTO^@BOOK

HALT

halt:postcondition

Ends the process. Releases all locked names, closes all 

opened devices, and aborts all active transactions. Halt never 

takes an argument.

>H

HANG

hang:postcondition  NumericExpression, . . .

Suspends execution of the process for approximately the 

specified number of seconds. Negative numbers or zero do 

not stop execution. Hang always takes an argument. [For 

portability, the effects of hanging for non-integer durations, e.g. 

2.5, should not be relied upon.]

>H 3

>H TIME

IF

if

if  TruthValueExpression, . . .

Lets the rest of the line execute only if all arguments evaluate 

to true; sets $test to whether the if succeeded. Argumentless 

if lets the rest of the line execute only if $test = 1. Note that 

because if with multiple arguments is identical to multiple, 

independent if commands, later arguments are evaluated 

only if earlier ones succeed.

>I

>I AUTHOR="CERVANTES"

>I FRIEND1="QUIXOTE",FRIEND2="SANCHO"

>I NAME="ALDONZA"!(NAME="DULCINEA")

>I $D(VIRTUE),VIRTUE="FRIENDS" D VIRTUE^CANON

JOB

job:postcondition  JobArgument:JobParams:timeout, . . .

Job Argument =>

entry reference

label reference(actual parameters)

Job Parameters =>

vendor-specific expression

(expression:expression: . . . )

Makes an independent process that begins execution at the 

specified line of code. Timed job sets $test to whether the job 

command succeeded in the time specified. Note that the job 

command can only pass parameters by value; trying to pass 

by reference causes error m40. [Implementation-specific JOB 

parameters are not portable.]

>J ^LISTPLAY

>J PRINT^CANON:25 E  Q

>J PRINT^ESSAY("MONTAIGNE","OF EXPERIENCE")

KILL

kill:postcondition

kill:postcondition  KillArgument, . . .

Kill Argument =>

local, global, or structured system variable name

(local variable name,local variable name, . . . )

Removes specified variables, and all their array descendants. 

Argumentless kill removes all local variables, and their array 

descendants. An exclusive kill, a kill argument in the form of 

a parenthesized list of local variables, removes all local 

variables and their descendants, except those listed within the 

parentheses and their descendants.

>K

>K PLAY,AUTHOR,^CANON("MOLIERE","TEMP")

>K (AUTHOR,WORK)

LOCK

lock:postcondition

lock:postcondition  SignLockArgument:timeout, . . .

Sign => + or -

LockArgument =>

Name Reference

(NameReference,NameReference, . . . )

Gets and/or releases ownership of names. Argumentless 

lock releases ownership of all names held by current process. 

Names, like devices, can only be owned by one process at a 

time. Ownership of a name includes all array descendants of 

that name.

Locked names are logical names that look like local and 

global names for convenience only. A name used as an 

argument of a lock command does not change the naked 

reference, nor can it be referred to by the naked reference.

Unlike devices, names can be used when not owned; 

locking is a voluntary signaling mechanism, not a method for 

preventing access. Locking a name does not prevent other 

processes from accessing or manipulating variables with the 

same names, only from successfully locking those names. 

Mumps code should voluntarily lock names, especially global 

names, whose simultaneous use by multiple processes would 

lead to problems, such as loss of database integrity. 

Each non-incremental lock argument (those without + or 

-) first releases all names owned by this process, and then 

gets ownership of the name specified. Incremental and 

decremental lock arguments get (+) or release (-) ownership 

of the specified names without releasing any other names. A 

name incrementally locked multiple times must be 

decrementally locked an equal number of times to release it.

A lock of a list of names enclosed by parentheses will not 

succeed until all names listed are simultaneously available. A 

nonincremental lock of a list of names will not release the 

names owned by the process until it can simultaneously get 

ownership of all the names in its list. 

Timed lock sets $test to whether it got ownership of the 

name in the time requested; network latency and other 

extraneous delays are not counted in the timeout period. For 

arguments without a timeout, lock waits indefinitely until the 

name is released by whatever process owns it. Improperly 

coded, this can result in a deadly embrace, in which two 

processes stop execution at lock commands because each 

owns names for which the other is waiting.

>L

>L ^CANON("MILTON")

>L (CRITIC,^CANON("JOHNSON"))

>L ^CANON("GOETHE"):2

>L +^CANON("AUSTEN")

>L -^CANON("WORDSWORTH")

MERGE

merge:postcondition  Variable=Variable, . . .

Variable =>

global, local, or structured system variable

Copies the value and all array descendants from one variable 

to another variable. If either variable is an array descendant of 

the other, it causes error m19.

>M WHITMAN=^CANON("WHITMAN")

NEW

new:postcondition

new:postcondition  NewArgument, . . .

New Argument =>

local variable name

(local variable name,local variable name, . . . )

$estack

$etrap

Saves and temporarily removes locals and their array 

descendants, and restores them when this block of code 

ends. Argumentless new saves and temporarily removes all 

locals and their array descendants, and restores them when 

this block of code ends. 

Only unsubscripted local variables or the intrinsic special 

variables $estack and $etrap may be used in the argument of 

the new command. An exclusive new, a new argument 

consisting of a list of names within parentheses, saves and 

temporarily removes all locals, except those listed and their 

descendants.

>N

>N DICKENS,ELIOT

>N (TOLSTOY,IBSEN)

OPEN

open:postcondition  Device:OpenParameters, . . .

Open Parameters =>

DeviceParameters:timeout:MnemonicSpecs

Device Parameters =>

Device Parameter

 (Device Parameter:Device Parameter: . . . )

Device Parameter =>

vendor-specific expression

Device Keyword

Device Attribute=expression

Mnemonic Specs =>

Mnemonic Space

(Mnemonic Space,Mnemonic Space, . . . )

Gets ownership of an i/o device, selects the list of available 

mnemonic spaces for that device, and sets the current 

mnemonic space to the first in the list selected. Open can also 

use device parameters to manipulate the device as it is 

acquired. Trying to open a device with a mnemonic space it 

doesn’t support causes error m35; trying to open it with 

incompatible mnemonic spaces causes error m36. Timed open 

sets $test to whether it got ownership in the specified time. 

[Implementation-specific OPEN parameters and nonstandard 

mnemonic spaces are not portable.]

>O LOGFILE,PRINTER::60 E  D FAIL Q

>O:AUTHOR="DICKINSON" "MIND":("NEW"):1

>O DISPLAY:::"X3.64"

QUIT

quit:postcondition

quit:postcondition  Expression

Ends the current process level and returns a value; doing so 

when an argument is not expected causes error m16. 

Argumentless quit ends the current process level without 

returning a value, but if one is expected it causes error m17. 

Quit can also be used to end a for loop on the same line.

>Q

>Q "FREUD"

>S AUTHOR="PROUST",NAME="AUTHOR" Q @NAME

READ

read:postcondition  ReadArgument, . . .

ReadArgument =>

String Literal

Formatting String

*Variable:timeout

Variable#ReadCount:timeout

Formatting String =>

FeedsTab

/ControlMnemonic(expression,expression, . . . )

Feeds => !s (line feeds) or #s (form feeds)

Tab => ? followed by column number

Variable =>

local, global, or structured system variable

ReadCount => integer expression

Gets input from the current i/o device and puts the response 

in the specified variables. Any text and format control 

characters in the argument of the read command are output 

on the current device. Timed read sets $test to whether read 

got a response in the specified time.

When the argument contains an asterisk preceding a 

variable name, a code representing a single character is 

obtained. When the argument contains a variable followed by 

a "#" and a numeric expression, this expression specifies the 

maximum number of characters to accept. A number less 

than zero causes error m18. [The codes returned by READ * are 

defined by the MUMPS implementor and not portable. Use of 

nonstandard control mnemonics is not portable.]

>R #!!,AUTHOR,!?5,WORK 

>R *CODE

>R !,NAME#10

>R "NAME:",AUTHOR G PORTRAIT:AUTHOR="JOYCE"

>R !?10,"WOOLF?",YES:30 I '$T W !,"AFRAID?"

SET

set:postcondition  SetDestination=Expression, . . .

Set Destination => SetLeft or (SetLeft,SetLeft, . . . )

Set Left =>

local, global, or structured system variable

$device or $key or $x or $y

$ecode or $etrap

$extract(string,from,to)

$piece(string,delimiter,from,to)

Puts values into variables. When the set destination is a list of 

destinations within parentheses, each destination is given the 

value following the assignment symbol. The $extract and 

$piece destinations change the specified part of their first 

arguments. Trying to set $x or $y to a negative or non-integer 

value causes error m43.

>S AUTHOR="KAFKA",STORY="METAMORPHOSIS"

>S AUTHOR="BORGES",^CANON(AUTHOR)=1

>S (AUTHOR,CURRENT,^CANON("ACTIVE"))="NERUDA"

>S $ET="TRAP^CANON",$EC=",U42,"

>S $P(^CANON("PESSOA"),"^",3)="MENSAGEM"

>S $E(^CANON("CRITIC"),1,20)=$J("BLOOM",20)

TCOMMIT

tcommit:postcondition

Commits and ends the current transaction: makes its global 

changes visible. A tcommit when there is no current 

transaction causes error m44.

>TC:AUTHENTC

TRESTART

trestart:postcondition

Rolls back the current transaction (see trollback), optionally 

restores some or all of the symbol table (as dictated by the 

tstart command, see below), and starts the current 

transaction again. Attempting to restart a non-restartable 

transaction rolls back the transaction, ends it, and causes error 

m27. A trestart when there is no current transaction causes 

error m44.

>TRE:MISSING

TROLLBACK

trollback:postcondition

Rolls back a transaction; that is, undoes its global changes and 

releases any locks acquired within the transaction. A 

trollback when there is no current transaction causes error 

m44.

>TRO:CANCEL

TSTART

tstart:postcondition

tstart:postcondition  Variables:TransParameters

Variables =>

local variable name

(local variable name,local variable name, . . . )

* or ()

Transaction Parameters =>

Parameter

(Parameter:Parameter: . . . )

Parameter =>

serial

transactionid=expression

z . . . =expression

Starts a restartable transaction. Argumentless tstart starts a 

nonrestartable transaction. When variables are specified, they 

are restored to their previous values if the transaction is 

restarted. [Use of transaction parameters starting with Z is not 

portable.]

>L ^CANON("UPDATE") TS

>TS:OK :T="BECKETT"

>TS (CHANGES,VALUES):(S:T="HOMER")

USE

use:postcond  Device:DeviceParams:MnemSpace, . . .

Device Parameters =>

Device Parameter

 (Device Parameter:Device Parameter: . . . )

Device Parameter =>

vendor-specific expression

Device Keyword

Device Attribute=expression

Picks the current device from the list of i/o devices owned by 

the current process, and the device’s mnemonic space from 

the list currently available for that device. Use can also use 

device parameters to manipulate the current device as it is 

selected. [Implementation-specific USE parameters and 

nonstandard mnemonic spaces are not portable.]

>U:AUTHOR="HERODOTUS" "LOG.TXT"

>U:AUTHOR="THUCYDIDES" HISTORY::"X3.64"

VIEW

view:postcondition  ViewArgument

Returns or changes implementation-dependent information. 

[Arguments and behavior of VIEW are nonstandard.]

>VIEW:AUTHOR="PLATO" MEMORY

WRITE

write:postcondition  WriteArgument, . . .

WriteArgument =>

Expression

Formatting String

*Character Code

Formatting String =>

FeedsTab

/ControlMnemonic(expression,expression, . . . )

Feeds => !s (line feeds) or #s (form feeds)

Tab => ? followed by column number

Variable =>

local, global, or structured system variable

Formats and outputs values to the current i/o device. When 

an argument includes an asterisk followed by an integer value, 

one character whose code (not necessarily ascii) is the 

number represented by the integer is sent to the current 

device; the effect this code has on the device is 

implementation-specific. [The codes used by WRITE * are 

defined by the MUMPS implementor and not portable. Use of 

nonstandard control mnemonics is not portable.]

>W "SOPHOCLES"

>W #!!?15,"EURIPIDES: ",PLAY

>W *7

>W !,^CANON("ACTIVE"),?30,WORK

XECUTE

xecute:postcondition  Expression:postcondition, . . .

Interprets and executes a value as mumps text. Xecute 

provides a means of interpreting a data value created during 

program execution as if it were mumps code. Each argument 

of the xecute command is interpreted as if it were the text 

part of a line of mumps code (without label, line start indicator, 

or line level indicator).

>X "S AUTHOR=""ARISTOTLE"""

Z .  .  .

All implementation-specific command names begin with z. 

[Commands whose names begin with Z are not portable.]

For more information on commands, see chapter 8 of MUMPS 

1995.

9  !  C H A R A C T E R  S E T  P R O F I L E S

Mumps systems can be assigned a default character set 

profile; individual processes, globals, devices, and routines 

can be assigned their own character sets. Structured system 

variables define these defaults and assignments.

9 .1  !  C H A R A C T E R  S E T S  A S C I I  &  M

Ascii and m operate upon the same set of characters—the 

seven-bit ascii character set—and use the same pattern code 

definitions, but differ in their collation algorithms.

Ascii collates in strict ascii numeric order applied lexically. 

That is, 1 collates before 2, but so do 12 and "1A", just as "A" 

collates before "AB" which collates before "B".

M collates the same except that canonical numbers 

collate numerically ahead of all other strings. To use the 

previous example, "A", "AB", and "B" will collate the same 

as they do in the ascii character set, but 2 collates ahead of 12, 

and all three numbers collate before the string "1A".

9.1 . 1  !  Table  of Ascii/M Char acters

# char codes

0 nul c,e

1 soh c,e

2 stx c,e

3 etx c,e

4 eot c,e

5 enq c,e

6 ack c,e

7 bell c,e

8 bs c,e

9 ht c,e

10 lf c,e

# char codes

11 vt c,e

12 ff c,e

13 cr c,e

14 so c,e

15 si c,e

16 dle c,e

17 dc1 c,e

18 dc2 c,e

19 dc3 c,e

20 dc4 c,e

21 nak c,e

# char codes

22 syn c,e

23 etb c,e

24 can c,e

25 em c,e

26 sub c,e

27 esc c,e

28 fs c,e

29 gs c,e

30 rs c,e

31 us c,e

32 sp p,e

# char codes

33 ! p,e

34 " p,e

35 # p,e

36 $ p,e

37 % p,e

38 & p,e

39 ' p,e

40 ( p,e

41 ) p,e

42 * p,e

43 + p,e

44 , p,e

45 - p,e

46 . p,e

47 / p,e

48 0 n,e

49 1 n,e

50 2 n,e

51 3 n,e

52 4 n,e

53 5 n,e

54 6 n,e

55 7 n,e

56 8 n,e

57 9 n,e

58 : p,e

59 ; p,e

60 < p,e

61 = p,e

62 > p,e

63 ? p,e

64 @ p,e

# char codes

65 A a,u,e

66 B a,u,e

67 C a,u,e

68 D a,u,e

69 E a,u,e

70 F a,u,e

71 G a,u,e

72 H a,u,e

73 I a,u,e

74 J a,u,e

75 K a,u,e

76 L a,u,e

77 M a,u,e

78 N a,u,e

79 O a,u,e

80 P a,u,e

81 Q a,u,e

82 R a,u,e

83 S a,u,e

84 T a,u,e

85 U a,u,e

86 V a,u,e

87 W a,u,e

88 X a,u,e

89 Y a,u,e

90 Z a,u,e

91 [ p,e

92 \ p,e

93 ] p,e

94 ^ p,e

95 _ p,e

96 ` p,e

# char codes

97 a a,l,e

98 b a,l,e

99 c a,l,e

100 d a,l,e

101 e a,l,e

102 f a,l,e

103 g a,l,e

104 h a,l,e

105 i a,l,e

106 j a,l,e

107 k a,l,e

108 l a,l,e

109 m a,l,e

110 n a,l,e

111 o a,l,e

112 p a,l,e

113 q a,l,e

114 r a,l,e

115 s a,l,e

116 t a,l,e

117 u a,l,e

118 v a,l,e

119 w a,l,e

120 x a,l,e

121 y a,l,e

122 z a,l,e

123 { p,e

124 | p,e

125 } p,e

126 ~ p,e

127 del c,e

9 . 2  !  C H A R A C T E R  S E T  J I S 9 0

Jis90, a character set for representing Japanese characters, is 

described in two separate standards: Japanese Industrial 

Standard JIS X0201-1990 8-bit coded character sets for information 

interchange, and Japanese Industrial Standard JIS X0208-1990 8-

bit double byte coded KANJI sets for information interchange.

MUMPS 1995 Annex H describes the relationship between 

these two standards within jis90, their pattern codes and 

collation, and which characters may be included within mumps 

names (variables, routines, labels, etc.).

3  COLOPHON 1

The first printing of this book was designed and set into type 

in 1995 in Seattle by the author in Bitstream’s ITC Friz Quadrata 

(Flareserif 861) and Monotype’s Courier New, using Microsoft 

Word version 6.0a on Intel 386 and 486 machines.

The author reset the second printing in 2010 using 

Neooffice 3.0 on a Macintosh Macbook Pro; his goals were 

(1) to update the software format, (2) to convert its styles to 

create a navigation outline for the portable-document format, 

and (3) to update the layout and design to be more readable, 

restful on the eyes, and typographically coherent.

The text face is Friz Quadrata version 2.00. Swiss designer 

Ernst Friz created it in 1965 as a solo Roman face for Visual 

Graphics Corporation. New York designer Victor Caruso 

added the bold face for International Typeface Corporation’s 

(itc) 1973 release. French designer Thierry Puyfoulhoux added 

the italic and bold italic faces for itc in 1992. Friz Quadrata is a 

spur-serif typeface legible enough for setting extended text; 

its clean lines and precision suit it to formal engineering texts 

like this pocket guide, but its open counters help breathe life 

and character into the text. The English text is set 8/14.

The code face is Courier 10 Pitch (Fixed 

Pitch 810) version 1.01, developed by Howard 

“Bud” Kettler in Lexington in 1955 for IBM 

typewriters and digitized by Bitstream in the 

1980s. Designed to give dignity to official 

documents, Courier has become the most famous 

monospaced typeface; most software is printed in 

Courier. Courier 10 Pitch’s color blends better 

with text faces than Courier New (which is far 

too light). The MUMPS text is set 8/14.

